
;-------------------------------------------------------------

; sl_3d.dat        influence of slope curvature on stability

;                  of an open pit

;-------------------------------------------------------------

; create model by joining radial cylinder and brick primitives

gen zone radcyl &

p0 0 0 25 p1 add 80 0 0 p2 add 0 0 -25 p3 add 0 80 0 &

dim 24.5 24.5 12 12 ratio 1 1 1 1.1

gen zone radcyl &

p0 0 0 0 p1 add 80 0 0 p2 add 0 0 -15 p3 add 0 80 0 &

dim 12 12 12 12 ratio 1 1 1 1.1 fill size 5 7 10 10

gen zone brick  &

p0 0 -40 -15 p1 add 12 0 0 p2 add 0 40 0 p3 add 0 0 15 &

size 5 12 7

gen zone brick  &

p0 12 -40 -15 p1 add 68 0 0 p2 add 0 40 0 p3 add 0 0 15 &

size 10 12 7 ratio 1.1 1 1

gen zone brick  &

p0 12 -40 0 p1 add 68 0 0 p2 add 0 40 0 p3 add 12.5 0 25 &

p4 add 68 40 0 p5 add 12.5 40 25 p6 add 68 0 25 p7 add 68 40 25 &

size 10 12 10 rat 1.1 1 1

; install water table

ini pp 0 grad 0 0 -1e4 range z -15.1 0

call water.fis

; assign Mohr Coulomb model and properties

mod mo

pro bulk 2e8 she 1e8 fric 45 coh 1e6 ten 1e6

; boundary conditions

fix x range x -.1 .1

fix x range x 79.9 80.1

fix y range y -40.1 -39.9

fix y range y 79.9 80.1

fix x range z -15.1 -14.9

fix y range z -15.1 -14.9

fix z range z -15.1 -14.9

; initialize density: saturated and unsaturated

def ini_dens

   pnt = zone_head

   loop while pnt # null

      if z_pp(pnt) # 0.0 then

         z_density(pnt) = 2600

      else

         z_density(pnt) = 2500

      endif



      pnt = z_next(pnt)

 endloop

end

ini_dens

set g 0 0 -10

; initialize gravity stresses

ini syy -6.25e5 grad 0 0 2.5e4

ini sxx -6.25e5 grad 0 0 2.5e4

ini szz -6.25e5 grad 0 0 2.5e4

; histories

hist unbal

hist gp xdisp 24.5 -40 25

hist gp zdisp 24.5 -40 25

hist gp xdisp 24.5 0 25

hist gp ydisp 24.5 0 25

hist gp zdisp 24.5 0 25

hist gp ydisp 0 24.5 25

hist gp zdisp 0 24.5 25

; equilibrate

solve

save sl_eq.sav

; reduce strength

pro coh 3.7e4

ini xd 0

ini yd 0

ini zd 0

ini xv 0

ini yv 0

ini zv 0

step 4000

save sl_fail.sav

;

; create plot views of results

;

; view : sl_geom  (model grid)

plot create sl_geom

set out sl_geom.plt

plot set rot 20 0 50

plot set ang 20

plot set cent 40 20 5

plot add surface lcyan

plot add axes brown

plot hard

;



; view : sl_water (water table with model grid)

plot create sl_water

set out sl_water.plt

plot set rot 20 0 50

plot set ang 20

plot set cent 40 20 5

plot add water lgra

plot add sketch black

plot hard

;

; view : sl_ppres  (pore pressure contours)

plot create sl_ppres

set out sl_ppres.plt

plot set rot 20 0 50

plot set ang 20

plot set cent 40 20 5

plot add cont pp int 2.5e4 sh on

plot hard

;

; view : sl_disp  (displacement contours)

plot create sl_disp

set out sl_disp.plt

plot set rot 20 0 50

plot set ang 20

plot set cent 40 20 5

plot add cont disp sh on

plot hard

;

; view : sl_hist  (displacement histories)

plot create sl_hist

set out sl_hist.plt

plot add hist 2 3 7 8

plot hard

ret

�


